CE 310

Excel Tutorial — Conditional Probability in Excel

Week 3 · Session 2 · Hands-on Walkthrough

University of Arizona · Dr. Wooyoung Jung · Fall 2026

Today's Plan — Four New Skills, Then the In-Class Exercise

By the end of this walkthrough you will know how to:

1. Navigate the new energy dataset and identify its **text columns**
2. Build an **IF helper column** to flag rows meeting a threshold
3. Use **COUNTIFS with text criteria** — `"Summer"`, `"Yes"`, `"Weekday"`
4. Use **AVERAGEIFS** for multi-condition averages (argument order differs from AVERAGEIF)
5. Compute a **conditional probability** as a COUNTIFS / COUNTIF ratio

Assumed from Weeks 1–2: AVERAGEIF, COUNTIFS with numeric criteria, histogram, PERCENTILE. Today the criteria are text strings — the syntax is slightly simpler, but the columns look different.

Step 1 — Open the New Dataset

File: `Week3_Tucson_Office_Energy_2023.csv` · 8,760 rows · 1 row per hour of 2023

Column	Header	Type
A	DateTime	Date/time
D	DayType	Text ("Weekday" / "Weekend")
E	Season	Text ("Winter/Spring/Summer/Fall")
F	BusinessHours	Text ("Yes" / "No")
H	Cooling_kWh	Number

> 💡 Columns D, E, and F contain **text strings**, not numbers. You will type these strings directly as criteria in COUNTIFS — no `&` operator needed. This is the key difference from Week 2.

Save as `.xlsx` immediately. Press **Ctrl + End** to verify the last row is **8761**. Use `=COUNTA(A:A)-1` to double-check the row count.

Step 1 — Text Columns, Not Numbers

Three text columns, two number columns

text — left-aligned by Excel   numbers — right-aligned

	D	E	F	G	H
1	DayType	Season	BusinessHours	Electricity_kWh	Cooling_kWh
3632	Weekday	Summer	No	76.49	22.91
3633	Weekday	Summer	No	162.52	38.49
3634	Weekday	Summer	Yes	162.33	46.27
3635	Weekday	Summer	Yes	169.85	55.31
⋮					
8761					

`=COUNTIFS(E2:E8761,"Summer")` the string on its own — no & , no operator

`=COUNTIFS(H2:H8761,">50")` a number still needs its operator inside the quotes

A misspelling returns 0 rather than an error — "Summmer" counts nothing and says nothing.

Step 2 — IF Helper Column: Flag High-Cooling Hours

A helper column marks each row 1 (condition met) or 0, so you can count it with `=SUM`.

> 📌 **In cell M1, type** `IsCoolHigh`
>
> **In M2, type:** `=IF(H2>50, 1, 0)`
> Press Enter → you should see **0** (January hour 1 has low cooling demand).
>
> Double-click the fill handle to fill M3 through M8761.
> Then in N1 type `IsElecHigh` and in N2 type `=IF(G2>100, 1, 0)`. Fill down.

`=IF(test, 1, 0)` returns 1 when true, 0 when false. (Use 1/0, not TRUE/FALSE — SUM ignores boolean cells.)

Verify: `=SUM(M2:M8761)` should equal `=COUNTIF(H2:H8761, ">50")` — two ways to count the same rows.

Step 2 — The Flag Column in Place

Step 2 — the helper column sits past the data

M2 | *fx* `=IF(H2>50, 1, 0)`

	H	I	J	K	L	M
1	**Cooling**	**HeatElec**	**Light**	**Equip**	**Gas**	**IsCoolHigh**
3632	22.9	1.0	16.1	32.7	1.1	**0**
3633	38.5	0.0	48.2	70.4	1.1	**0**
3634	46.3	0.0	48.2	62.0	1.1	**0**
3635	55.3	0.0	48.2	60.1	1.1	**1**
⋮						
8761						

Each row gets a 1 or a 0. Two ways to count the same rows, and they must agree:

`=SUM(M2:M8761)` = **`=COUNTIF(H2:H8761,">50")`** both give 1,245

Use 1 and 0, not TRUE and FALSE — SUM ignores boolean cells.

Step 3 — COUNTIFS with Text Criteria

In Week 2, criteria joined an operator with a cell value: `">="&I2`. For text columns, write the string **directly in quotes** — no `&` needed.

📌 **Type these in blank cells and verify the counts:**

`=COUNTIFS(E2:E8761, "Summer")` → Summer hours ≈ **2,208**
`=COUNTIFS(F2:F8761, "Yes")` → Business-hours rows ≈ **2,600**
`=COUNTIFS(D2:D8761, "Weekday")` → Weekday hours ≈ **6,240**

How the count works: Summer = June (30) + July (31) + August (31) = 92 days × 24 hours = **2,208**. BusinessHours "Yes" = Mon–Fri 8am–6pm = 10 hours × 260 weekdays = **2,600**.

⚠️ Text criteria are case-insensitive ("summer" works) but must be spelled exactly right. A typo like "Summmer" returns **0 with no error message**. If you get 0 when you expect thousands, check spelling in both the formula and the Season column header row.

Step 4 — COUNTIFS with Text and Number Criteria

COUNTIFS can mix text and number conditions freely. Each condition pair is tested independently; all must be true simultaneously.

```
=COUNTIFS(E2:E8761, "Summer", H2:H8761, ">50")
```

📌 **Type these and verify:**

Formula	What it counts	Result
`=COUNTIFS(E2:E8761, "Summer", H2:H8761, ">50")`	Summer AND Cooling > 50 kWh	≈ 835
`=COUNTIFS(E2:E8761, "Summer", F2:F8761, "Yes")`	Summer AND business hours	≈ 660
`=COUNTIFS(F2:F8761, "Yes", G2:G8761, ">140")`	Business hours AND Electricity > 140 kWh	≈ 1,539

Step 5 — AVERAGEIFS: Average by Multiple Conditions

`=AVERAGEIF` (Week 1) takes **one** condition. `=AVERAGEIFS` takes **two or more** — but the argument order is **reversed**.

⚠️ **Argument order is opposite between the two functions:**

`=AVERAGEIF(crit_range, crit, avg_range)` — avg_range is **last** (3rd arg)

`=AVERAGEIFS(avg_range, crit_range, crit, ...)` — avg_range is **first** (1st arg)

📌 **Type these and confirm:**

`=AVERAGEIFS(G2:G8761, F2:F8761, "Yes")`

→ avg electricity during business hours ≈ **152 kWh**

`=AVERAGEIFS(H2:H8761, E2:E8761, "Summer", F2:F8761, "Yes")`

→ avg cooling: Summer + business hours ≈ **74.5 kWh**

Step 5 — The Argument Order Reverses

AVERAGEIF and AVERAGEIFS take the same arguments in opposite order

AVERAGEIF

`=AVERAGEIF(crit_range, crit, avg_range)` **the averaged column is LAST**

same thing, moved to the front

AVERAGEIFS

`=AVERAGEIFS(avg_range, crit_range, crit, …)` **the averaged column is FIRST**

In practice, on this week's data:

`=AVERAGEIF(F2:F8761,"Yes",G2:G8761)` one condition — business hours

`=AVERAGEIFS(H2:H8761,E2:E8761,"Summer",F2:F8761,"Yes")` two conditions — summer AND business hours

Swap them and Excel averages the criteria column — it does not warn you, because the request is legal.

Step 6 — Conditional Probability: The Formula Pattern

$$P(A \mid B) = \frac{\text{Count}(A \text{ AND } B)}{\text{Count}(B)}$$

In Excel: `= COUNTIFS(A_range, A_crit, B_range, B_crit) / COUNTIF(B_range, B_crit)`

Denominator = count of the conditioning group (all Summer hours, regardless of cooling).

Numerator = count of the joint event (Summer hours that ALSO exceed the cooling threshold).

> 📌 **Compute P(Cooling > 50 kWh | Season = Summer):**
>
> Denominator: `=COUNTIFS(E2:E8761, "Summer")` → ≈ 2,208
>
> Numerator: `=COUNTIFS(E2:E8761, "Summer", H2:H8761, ">50")` → ≈ 835
>
> Probability: `= 835 / 2208` → ≈ **37.82%**

💡

Every in-class exercise cell follows the same pattern: COUNTIFS ÷ COUNTIF. Only the column and threshold change.

Step 6 — Which Group Is the Denominator

A conditional probability is a choice of denominator

$$P(\text{Cooling} > 50 \mid \text{Summer}) = \frac{835}{2{,}208} = \mathbf{37.82\%}$$

The denominator is the group you are conditioning on — not the whole year.

Step 7 — What Conditioning Reveals

The same threshold (Cooling > 50 kWh) gives very different probabilities depending on context:

Question	Denominator	Probability
P(Cooling > 50) — all hours	8,760 (full year)	≈ **14.21%**
P(Cooling > 50 | Summer)	≈ 2,208 (summer hours only)	≈ **37.82%**
P(Cooling > 50 | January)	744 (January hours)	≈ **0%**

💡 The marginal 14.21% is a weighted average across all seasons — including cold nights and winter months when cooling never runs. Conditioning on Summer reveals that nearly **38% of summer hours** require high cooling capacity. A system sized for the annual average would be undersized for nearly 4 out of every 10 summer hours.

Step 7 — Same Threshold, Three Answers



The marginal 14.21 % averages over seasons that never cool. Size for it and nearly four in ten summer hours fall short of capacity.

Pivot Table — For B11 and B12 in the In-Class Exercise

The in-class exercise asks for P(Cooling > 50 | Season) and then the same thing for all twelve months. COUNTIFS does it one cell at a time; a pivot table does the whole breakdown at once.

📌 Click any cell in the data → **Insert** → **PivotTable** → **New Worksheet** → OK.

Drag **Season** into **Rows** and **IsCoolHigh** into **Values**. Then click the field in Values → **Value Field Settings** → change **Sum** to **Average**.

💡 The average of a 0/1 column *is* the proportion of 1s — which is the conditional probability. That is the only reason this works.

Pivot Table — The One Setting That Decides the Answer

PivotTable — and the dropdown that decides what you get

Click any cell in the data first, so Excel proposes the whole range on its own.

1 Insert → PivotTable

Create PivotTable ×

Table/Range:

`Data!$A$1:$M$8761`

Choose where to place the report:

- ◉ **New Worksheet**
- ○ Existing Worksheet

OK Cancel

2 Drag two fields

PivotTable Fields ×

- ☐ DateTime
- ☑ **Season**
- ☐ Cooling_kWh
- ☑ **IsCoolHigh**

Rows

Season

Values

Average of IsCoolHigh

3 Value Field Settings

Values → dropdown →
Value Field Settings...

Summarize values by

- Sum
- Count
- **Average**
- Max
- Min

Sum is what Excel picks for you.
Average is what the question asks for.

✗ **Sum of IsCoolHigh, Summer row → 835. That is a count of hours, and it is not a probability.**

✓ **Average of IsCoolHigh, Summer row → 0.3782, i.e. 37.8%. The mean of a 0/1 column is a proportion.**

⚠ **The pivot lands on a sheet Excel names itself — copy the twelve monthly values into Work!B7:B18.**

Charting the Monthly Profile

Chart 1 plots the twelve monthly probabilities with a **reference line at your A3 marginal value**, and asks for **data labels** on the bars.

Build it from the `Work` sheet, not from the pivot chart — a pivot chart will not take an extra series cleanly.

> 📌 Copy the twelve monthly values from the pivot into `Work!B7:B18` (× 100 for percentages).
>
> 1. `C6` — header `A3 marginal %`; `C7` — your A3 value; fill down to `C18`
> 2. Select `A6:C18` → **Insert** → **Charts** → **Column**
> 3. Chart Design → Add Chart Element → **Data Labels**
> 4. Right-click the flat series → **Format Data Series** → dashed line

> 💡 Same move as Week 1: a reference line is a **column holding the same number over and over**. Never draw one with *Insert* → *Shapes* — it will not move when your A3 does.

Quick Check — Before You Start the In-Class Exercise

Think, then check

You computed:

- $P(\text{Cooling} > 50) \approx 14.21\%$
- $P(\text{Cooling} > 50 \mid \text{Summer}) \approx 37.82\%$

Question A: Write the COUNTIFS formula for P(Electricity > 100 kWh | BusinessHours = "Yes"). What is the denominator formula?

Question B: If you accidentally used `=AVERAGEIF` with AVERAGEIFS's argument order, where would the formula go wrong?

> Compare your answers with your neighbor. We'll confirm with the data.

The Step Every In-Class Exercise Ends With

Getting a finished chart into the box it is graded in

Every week. Build the chart on the sheet its data lives on, then move it onto the Charts sheet.

1 Right-click the chart



2 Choose “Object in”



3 It lands in the box



✗ **New sheet — makes a separate sheet called Chart1. The box on Charts stays empty and the points are lost.**

✓ **Object in → Charts — then drag the chart inside the outlined box.**

⚠ **Leave it a live chart — a pasted picture of one scores zero.**

Quick Reference — Week 3 New Formulas

Task	Formula
Count by text	`=COUNTIFS(E2:E8761, "Summer")`
Count text + number	`=COUNTIFS(E2:E8761,"Summer", H2:H8761,">50")`
IF flag column	`=IF(H2>50, 1, 0)` → fill down
Avg by 1 condition	`=AVERAGEIF(F2:F8761,"Yes", G2:G8761)`
Avg by 2 conditions	`=AVERAGEIFS(H2:H8761, E2:E8761,"Summer", F2:F8761,"Yes")`

Task	Formula
Marginal probability	`=COUNTIF(H2:H8761,">50") / 8760`
Denominator (group)	`=COUNTIFS(E2:E8761, "Summer")`
Numerator (joint)	`=COUNTIFS(E2:E8761,"Summer", H2:H8761,">50")`
Conditional probability	`= numerator / denominator`

Before You Submit

> ⚠️ Following this submission format exactly is essential — with a class this size, grading depends on it. Submissions that don't comply will be penalized.

1. **File format — `.xlsx` only.** Save → Save As → Excel Workbook (*.xlsx). Files saved as `.xls`, `.csv`, or `.pdf` cannot be graded.
2. **Sheet name — `Answers`.** Do not rename or delete this sheet.
3. **Column layout — do not change.** Answers go in Column E. Do not insert, delete, or move columns.
4. **File name — `CE310_W3_<NetID>.xlsx`** (replace `<NetID>` with your UA NetID).
5. **Upload** the `.xlsx` file to D2L Dropbox by **9:00 AM next Wednesday**.

You're Ready — Open the In-Class Exercise Instructions

`Wk03.03_In_Class_Exercise.pdf` · **`Wk03.04_In_Class_Answer_Sheet.xlsx`**

Your setup steps before Section A:

- Save as `.xlsx` · Freeze top row · Verify 8,760 rows (Ctrl + End)
- Column M: `IsCoolHigh` = `=IF(H2>50, 1, 0)` → fill to M8761
- Column N: `IsElecHigh` = `=IF(G2>100, 1, 0)` → fill to N8761

Take-home: B7–B14 (business-hours conditioning, pivot table, peak month, reverse conditioning), Chart 2, and the memo — typed into the `Written` sheet of your Answer Sheet, not a separate file.

Raise your hand if COUNTIFS returns 0 when you expect a large count — the most common cause is a spelling mismatch between your formula string and the actual text in the Season or DayType column.