CE 310

Excel Tutorial — Normal Distribution in Excel

Week 5 · Session 2 · Concrete QC, Traffic Speed, and Construction Cost

University of Arizona · Dr. Wooyoung Jung · Fall 2026

Today's Plan — Seven Skills

By the end of this tutorial you will be able to:

1. Navigate the new CE dataset and verify structure with `COUNTA`
2. Compare two σ values — concrete strength vs. traffic speed
3. Compute $P(X \le x)$ using `NORM.DIST` with `TRUE`
4. Compute $P(X > x)$ and $P(a < X < b)$ — exceedance and range patterns
5. Find design values using `NORM.INV` (ACI 318 + 85th percentile speed)
6. Compute conditional σ by Supplier with `STDEV(FILTER(...))` or array formula
7. Build a LN helper column and use `LOGNORM.DIST` for skewed costs

Step 1 — Open the Dataset

File: `Week5_CE_Measurements.csv` — 500 rows of CE field measurements

Column	Header	Content
A	RecordID	1–500, unique row identifier
B	Supplier	"Alpha" or "Beta" — concrete supplier
C	TimeOfDay	"AM_Peak" or "PM_Peak" — traffic observation period
D	fc_psi	Concrete compressive strength (psi)
E	Speed_mph	Traffic spot speed (mph)
F	CostPerCY	Unit cost of concrete placement ($/CY)

Step 1 (cont.) — Set Up and Verify

Setup: File → Save As → .xlsx · View → Freeze Top Row

Verify row count: `=COUNTA(A:A)-1` → should return **500**

(COUNTA counts all non-empty cells in column A, including the header — subtract 1 for the data rows.)

Col D is concrete strength — what the plant delivers. Col E is traffic speed — what drivers choose. Both are approximately Normal but with very different σ values. Col F is right-skewed — a different model applies.

Step 2 — Compare Two σ Values

In a new sheet (or cells H1:H8), compute descriptive statistics for both variables:

```
=AVERAGE(D2:D501)   → μ_fc    = 4,496 psi
=STDEV(D2:D501)     → σ_fc    =   442 psi
=MEDIAN(D2:D501)    → check: should be close to mean (symmetric)
=SKEW(D2:D501)      → should be near 0
=AVERAGE(E2:E501)   → μ_speed =  54.4 mph
=STDEV(E2:E501)     → σ_speed =   7.8 mph
```

Variable	μ	σ	Coefficient of Variation (σ/μ)
fc_psi	4,496 psi	442 psi	9.8%
Speed_mph	54.4 mph	7.8 mph	14.3%

Same NORM.DIST/NORM.INV tool for both engineering quantities — only μ and σ differ.

Step 3 — NORM.DIST: P(X ≤ x) — The CDF

Syntax: `=NORM.DIST(x, μ, σ, TRUE)` → returns P(X ≤ x) as a decimal

TRUE vs FALSE: The fourth argument is critical.

- `TRUE` → cumulative probability P(X ≤ x) — **always use this for probabilities**
- `FALSE` → probability density (height of the bell curve) — **not a probability**

If your answer looks like 0.000481 instead of 0.131, you used FALSE. Switch to TRUE.

Step 3 — NORM.DIST: $P(X \leq x)$ — The CDF (continued)

Compute P(fc < 4,000 psi) for all specimens (μ = 4,496, σ = 442):

```
=NORM.DIST(4000, 4496, 442, TRUE)*100   → 13.1%
```

Interpretation: 13.1% of all specimens — across both suppliers combined — fall below the 4,000 psi ACI specification.

This 13.1% mixes both suppliers. Step 6 separates Alpha from Beta — failure rates diverge sharply.

Step 3 — A Height Is Not a Probability



Step 4 — Three Probability Patterns

Every Normal probability question reduces to one of three patterns:

Pattern	Excel formula
Left tail: $P(X \le x)$	`=NORM.DIST(x, μ, σ, TRUE)`
Right tail: $P(X > x)$	`=1−NORM.DIST(x, μ, σ, TRUE)`
Between: $P(a < X < b)$	`=NORM.DIST(b,μ,σ,TRUE)−NORM.DIST(a,μ,σ,TRUE)`

Compute all four of the following using all-specimen parameters where applicable:

```
=NORM.DIST(4000, 4496, 442, TRUE)*100        → 13.1%  P(fc < 4,000)    [left]
=(1−NORM.DIST(5000, 4496, 442, TRUE))*100    → P(fc > 5,000 psi)      [right]
=(NORM.DIST(5000,4496,442,TRUE)−NORM.DIST(4200,4496,442,TRUE))*100
    → P(4,200 < fc < 5,000 psi)                                     [between]
=(1−NORM.DIST(45, 54.4, 7.8, TRUE))*100      → P(Speed > 45), all   [right]
```

Step 4 — The Same Curve, Shaded Three Ways



Step 5 — NORM.INV: From Probability to Design Value

Syntax: `=NORM.INV(p, μ, σ)` → returns x such that $P(X \leq x) = p$

NORM.INV is NORM.DIST in reverse: give it a probability, get back the design value at that percentile.

ACI 318 — 10th percentile compliance check:

```
=NORM.INV(0.10, 4632, 344)  → 4,191 psi   Supplier Alpha: ≥ 4,000 ✓ compliant
=NORM.INV(0.10, 4367, 484)  → 3,747 psi   Supplier Beta:  < 4,000 ✗ non-compliant
```

Traffic 85th percentile design speed:

```
=NORM.INV(0.85, 52.2, 8.2)  → 60.7 mph   AM Peak
=NORM.INV(0.85, 57.2, 6.2)  → 63.6 mph   PM Peak
```

p is a decimal, not a percent. `=NORM.INV(10, 4632, 344)` → #NUM! error. Use `0.10`, not `10`.

Step 5 — Reading the Curve in Reverse



Alpha's 10th percentile lands at 4,191 psi — above the specification, so Alpha is compliant.

Step 6 — Conditional σ by Supplier (Alpha vs Beta)

Excel has `AVERAGEIF` but no `STDEVIFS`. Two methods to compute conditional standard deviation:

Method 1 — Excel 365 (FILTER function):

```
=STDEV(FILTER(D2:D501, B2:B501="Alpha"))  → σ_Alpha ≈ 344 psi
=STDEV(FILTER(D2:D501, B2:B501="Beta"))   → σ_Beta  ≈ 484 psi
```

Method 2 — Array formula (all Excel versions):

```
=STDEV(IF(B2:B501="Alpha", D2:D501))   → press Ctrl+Shift+Enter
=STDEV(IF(B2:B501="Beta",  D2:D501))   → press Ctrl+Shift+Enter
```

Step 6 — Conditional σ by Supplier (continued)

Also compute conditional means:

```
=AVERAGEIF(B2:B501, "Alpha", D2:D501)  → μ_Alpha ≈ 4,632 psi
=AVERAGEIF(B2:B501, "Beta",  D2:D501)  → μ_Beta  ≈ 4,367 psi
```

Alpha's σ (344 psi) is 29% tighter than Beta's (484 psi). Same material, same 4,000 psi specification — different quality control. The narrower distribution is what makes Alpha compliant and Beta not.

Step 6 — What σ Buys You



Beta's mean is only 6 % lower than Alpha's. The failure rate is nearly seven times higher, and the width of the curve is the reason.

Step 7 — Lognormal Preview: LN Helper Column

CostPerCY is right-skewed (SKEW ≈ +1.5, mean \$195 > median \$185) — the lognormal model applies, not the Normal.

Why: if ln(Y) is bell-shaped, then Y is lognormal. Build a helper column to check.

G1: Type the header `LN_Cost`

G2: `=IF(F2>0, LN(F2), "")` — `LN` is the natural logarithm — then fill down to G501

Compute lognormal parameters from the helper column:

```
μ_ln:  =AVERAGE(G2:G501)  → ≈ 5.22
σ_ln:  =STDEV(G2:G501)    → ≈ 0.35
```

μ_ln comes from AVERAGE of LN(Cost), not LN(AVERAGE(Cost)). `LN(AVERAGE(F2:F501))` ≈ 5.27 — incorrect. Always transform first, then average.

Step 7 — Lognormal Preview: LN Helper Column (continued)

P(Cost > $250/CY):

```
=(1-LOGNORM.DIST(250, 5.22, 0.35, TRUE))*100  → ≈ 16%
```

Step 7 — Why the Logarithm Helps



The Three Charts This Week

Two histograms and a bar chart — 9 points. The histogram technique is **Week 2's**; you have not used it since.

> 📌 **Histograms (Charts 1 and 2).** In two blank columns build a bin table: `Bin_Low` and `Count`. At least 10 bins spanning the data range.
>
> `=COUNTIFS(D$2:D$501,">="&H2,D$2:D$501,"<"&(H2+250))` for `fc_psi` in 250 psi steps, filled down.
>
> Select both columns → **Insert** → **Charts** → **Column** → right-click a bar → **Format Data Series** → **Gap Width: 0**. Touching bars are what makes it a histogram rather than a bar chart.

> 📌 **Chart 3 — the 10% ACI threshold.** Add a third column holding `10` in both rows and include it in the selection — the same reference-line move as Week 1. A labelled **Insert** → **Text Box** is also accepted.

Quick Check — Before the In-Class Exercise

Work individually · 3 minutes

Question A: Write the Excel formula for P(4,200 < fc < 5,000 psi) using all-specimen parameters (μ = 4,496, σ = 442). Which of the three patterns does this use?

Question B: A student types `=NORM.INV(10, 4632, 344)`. What is wrong, and what should the formula be?

Question C: Supplier Beta's ACI 318 result: `=NORM.INV(0.10, 4367, 484)` = **3,747 psi**. Is Beta compliant with ACI 318? State two reasons from the data.

> Write your answers before looking at the next slide.

The Step Every In-Class Exercise Ends With

Getting a finished chart into the box it is graded in

Every week. Build the chart on the sheet its data lives on, then move it onto the Charts sheet.

1 Right-click the chart



2 Choose “Object in”



3 It lands in the box



✗ **New sheet — makes a separate sheet called Chart1. The box on Charts stays empty and the points are lost.**

✓ **Object in → Charts — then drag the chart inside the outlined box.**

⚠ **Leave it a live chart — a pasted picture of one scores zero.**

Quick Reference — NORM Functions

Function	Syntax	Returns	Use when
`NORM.DIST`	`(x, μ, σ, TRUE)`	$P(X \le x)$	You know x, want probability
`NORM.DIST`	`(x, μ, σ, FALSE)`	PDF density at x	Rarely needed in practice
`NORM.INV`	`(p, μ, σ)`	x such that $P(X \le x) = p$	You know p, want design value
`LOGNORM.DIST`	`(x, μ_ln, σ_ln, TRUE)`	$P(Y \le x)$ for lognormal Y	Right-skewed, always-positive data

Quick Reference — NORM Functions (continued)

Three probability patterns — the complete toolkit:

Question type	Formula
$P(X < a)$ — left tail	`=NORM.DIST(a, μ, σ, TRUE)`
$P(X > a)$ — right tail	`=1−NORM.DIST(a, μ, σ, TRUE)`
$P(a < X < b)$ — between	`=NORM.DIST(b,μ,σ,TRUE)−NORM.DIST(a,μ,σ,TRUE)`

Always use TRUE for probability questions. FALSE returns a density value that is not a probability.

Quick Reference — Supporting Formulas

Conditional statistics (no STDEVIFS in Excel):

Goal	Method 365	Method (all versions)
Conditional mean	`=AVERAGEIF(B:B, "Alpha", D:D)`	same
Conditional count	`=COUNTIF(B:B, "Alpha")`	same
Conditional σ	`=STDEV(FILTER(D2:D501, B2:B501="Alpha"))`	`=STDEV(IF(B2:B501="Alpha", D2:D501))` + Ctrl+Shift+Enter

Quick Reference — Supporting Formulas (continued)

Lognormal workflow — always transform first:

Step	Formula	Result
Helper column	`=IF(F2>0, LN(F2), "")` fill to G501	LN_Cost column
μ_ln	`=AVERAGE(G2:G501)`	≈ 5.22
σ_ln	`=STDEV(G2:G501)`	≈ 0.35
$P(Y > 250)$	`=(1-LOGNORM.DIST(250,5.22,0.35,TRUE))*100`	≈ 16%

Sanity checks: `=COUNTA(A:A)-1` → 500 · `=NORM.DIST(μ, μ, σ, TRUE)` → 0.50 · `=NORM.INV(0.50, μ, σ)` → μ

Before You Submit

File format and naming:

1. File must be saved as `.xlsx` (not .csv, .xls, or .xlsm)
2. File name: `CE310_W5_<NetID>.xlsx`, e.g., `CE310_W5_abc123.xlsx` — use your UA NetID, not your name
3. Required sheets: **Answers** and **Written** — do not rename or delete either tab
4. Column layout: keep all original columns (A–F) in place; add helper columns to the right (G onward)

Before You Submit (continued)

Before clicking Submit in D2L, verify:

- [ ] `=COUNTA(A:A)-1` returns **500** on your Answers sheet
- [ ] All NORM.DIST formulas use **TRUE** as the fourth argument
- [ ] NORM.INV probabilities are decimals: **0.10**, not **10**
- [ ] σ_Alpha ≈ **344** and σ_Beta ≈ **484** (conditional σ check)
- [ ] μ_ln from `AVERAGE(G:G)` ≈ **5.22** (lognormal parameter check)
- [ ] File name follows the pattern: `CE310_W5_<NetID>.xlsx`

You're Ready — Open the In-Class Exercise

`Week5_CE_Measurements.csv` · `Wk05.04_In_Class_Answer_Sheet.xlsx`

Work through the seven steps in order.

Complete Setup + Steps 1–5 (all NORM.DIST and NORM.INV problems) during class.

Steps 6 and 7 (conditional σ and lognormal) are the take-home portion.

Questions during the in-class exercise? Raise your hand — or post to the Week 5 discussion board on D2L.

Submission deadline: 9:00 AM next Wednesday, the start of class — same as every week.