CE 310

Excel Tutorial — Empirical Percentiles

Week 6 · Session 2 · Concrete, Acoustics, Traffic, and Cost

University of Arizona · Dr. Wooyoung Jung · Fall 2026

Step 1 — Verify the Datasets

File 1: `Week5_CE_Measurements.csv` — the same file from Week 5. No helper columns needed — every formula acts directly on the raw columns.

Column	Contents
A `RecordID`	1–500
B `Supplier`	"Alpha"/"Beta" (concrete)
C `TimeOfDay`	"AM_Peak"/"PM_Peak" (traffic)
D `fc_psi`	Concrete strength
E `Speed_mph`	Vehicle spot speed
F `CostPerCY`	Unit construction cost

Step 1 (cont.) — The New Noise File

File 2 (new): `Week6_Noise_Measurements.csv` — `A RecordID`, `B Location` ("Roadside"/"Courtyard"), `C NoiseLevel_dB`.

Verify row count on each file: `=COUNTA(A:A)-1` → should return **500** for both.

Step 2 — `LARGE` / `SMALL`: Exact-Rank Percentiles

`=LARGE(range, k)` = kth largest. `=SMALL(range, k)` = kth smallest. **No sorting, no distribution fitting.**

> **k for a percentile, n = 500:**
> `k = ROUND(n × p, 0)` counting from the bottom; `k = ROUND(n × (1−p), 0)` counting from the top.

Target	Formula	k	Result
10th percentile, all concrete	`=SMALL(D2:D501, 50)`	50	**3,930 psi**
90th percentile, all concrete	`=LARGE(D2:D501, 50)`	50	**5,053 psi**
85th percentile, all speed	`=LARGE(E2:E501, 75)`	75	**62.1 mph**
L10, noise file col C	`=LARGE(C2:C501, 50)`	50	**67.7 dB**

Step 2 — Counting From Which End



Step 3 — `PERCENTILE`: Interpolated Values

`=PERCENTILE(array, p)` returns the value at fractional position `p` (0 to 1), **linearly interpolating** between the two nearest sorted values.

Comparison	Formula	Result
10th percentile, all concrete (exact rank)	`=SMALL(D2:D501, 50)`	3,930 psi
10th percentile, all concrete (interpolated)	`=PERCENTILE(D2:D501, 0.10)`	**3,937 psi**
10th percentile, Supplier Alpha only (n=243)	`=PERCENTILE(FILTER(D2:D501,B2:B501="Alpha"), 0.10)`	**4,186 psi**

⚠️ ACI 318's own compliance rule is written as a fixed **fraction of tests**, so exact-rank `SMALL` / `LARGE` is the standard's own method — use it for compliance checks. Use `PERCENTILE` when a smooth interpolated estimate is what you need instead.

Step 3 — On a Value, or Between Two



Step 4 — Exceedance Probability with `COUNTIF`

$$P(X > x) = \frac{\text{COUNTIF(range, ">x")}}{n} \times 100\%$$

Question	Formula	Result
P(concrete < 4,000 psi)	`=COUNTIF(D2:D501,"<4000")/500*100`	**12.600%**
P(speed > 45 mph)	`=COUNTIF(E2:E501,">45")/500*100`	**88.200%**
P(cost > $250/CY)	`=COUNTIF(F2:F501,">250")/500*100`	**16.800%**

💡 Always divide by **500** (the total record count), not by the count returned from a different COUNTIF — a common copy-paste error when reusing a formula across questions.

Step 5 — Building a Duration Curve

Goal: rank all 500 concrete strength values from highest to lowest and plot them.

1. In column **H**, header `Rank`; H2 = 1, H3 = 2 … fill down to H501.
2. In column **I**, header `fc_Sorted`; `I2: =LARGE($D$2:$D$501, H2)`, filled down to I501.
3. Verify: I2 (rank 1) = highest strength tested; I501 (rank 500) = lowest.
4. In column **J**, header `ACI 318`; `J2` = your A2 answer (≈ 3,930), filled down all 500 rows — the same number every row.
5. Select `H1:J501` → **Insert** → **Charts** → **Line**.

Excel draws **two series**, because the selection had two value columns. The flat one is your reference line.

> 💡 A reference line is a **column holding the same number over and over** — never *Insert* → *Shapes*, which is attached to nothing.

Step 5 — What You Are Building



Step 5b — Chart 2 Is a Scatter, Not a Line

Chart 2 compares your **empirical** percentiles against the **Normal model** at the same 99 points. Both are curves through continuous values, so the chart type changes:

1. On the `Work` sheet, fill `B7:B105` (empirical) and `C7:C105` (model).
2. Select `A6:C105` — the percentile column and both value columns.
3. **Insert** → **Charts** → **Insert Scatter (X, Y)** → **Scatter with Smooth Lines**.
4. **Chart Design** → **Add Chart Element** → **Legend** so a reader can tell the two curves apart. Excel usually adds one as soon as there are two series.

Why scatter and not line. A line chart treats the x-axis as evenly spaced categories. Here x is a *number* — the percentile — and the spacing carries meaning, so the x-values have to be plotted as values. That is what Scatter (X, Y) does.

⚠️ Dots instead of curves? Right-click the series → **Change Chart Type** → the smooth-lines variant.

Quick Check — Before You Start the In-Class Exercise

Question A: What k would you use in `SMALL(E2:E501, k)` to find the 15th percentile of speed (n = 500)?

Question B: Supplier Beta's empirical 10th-percentile strength (3,688 psi) is noticeably below the Normal-model estimate from Week 5 (3,747 psi). Which supplier — Alpha or Beta — do you expect to show a *smaller* gap, and why?

Step 6 — Conditional Percentiles with `FILTER`

Same syntax as Week 5's conditional means/standard deviations — filter first, then apply `PERCENTILE`, `LARGE`, or `COUNTIF`.

Question	Formula	Result
85th percentile speed, AM Peak	`=PERCENTILE(FILTER(E2:E501,C2:C501="AM_Peak"),0.85)`	**60.90 mph**
85th percentile speed, PM Peak	`=PERCENTILE(FILTER(E2:E501,C2:C501="PM_Peak"),0.85)`	**62.78 mph**
Count of AM Peak records	`=COUNTIF(C2:C501,"AM_Peak")`	**278**

Array-formula fallback (pre-365 Excel): `{=PERCENTILE(IF(C2:C501="AM_Peak",E2:E501),0.85)}` — enter with **Ctrl+Shift+Enter**, not just Enter.

Step 6 — Two Populations, Two Answers



Step 7 — `SUMPRODUCT`: Total Exposure Above a Threshold

$$\text{Total excess} = \sum \max(\text{Cost} - \text{threshold},\ 0)$$

`=SUMPRODUCT((F2:F501-250)*(F2:F501>250))` → **$5,107.95**

How it works:

`(F2:F501-250)` computes the excess above $250 for every record (negative for records below the threshold).

`(F2:F501>250)` returns an array of TRUE/FALSE, which Excel treats as 1/0 — multiplying by this zeroes out every record that shouldn't count at all.

Step 7 — Only the Part Above the Line



The Step Every In-Class Exercise Ends With

Getting a finished chart into the box it is graded in

Every week. Build the chart on the sheet its data lives on, then move it onto the Charts sheet.

1 Right-click the chart



2 Choose “Object in”



3 It lands in the box



✗ New sheet — makes a separate sheet called Chart1. The box on Charts stays empty and the points are lost.

✓ Object in → Charts — then drag the chart inside the outlined box.

⚠ Leave it a live chart — a pasted picture of one scores zero.

Quick Reference — All Formulas This Week

Concept	Formula
kth largest / smallest	`=LARGE(range, k)` / `=SMALL(range, k)`
Interpolated percentile	`=PERCENTILE(range, p)`
Exceedance probability	`=COUNTIF(range, ">x") / n * 100`
Conditional percentile	`=PERCENTILE(FILTER(range, condition), p)`
IQR	`=PERCENTILE(range,0.75) - PERCENTILE(range,0.25)`
Total exposure above threshold	`=SUMPRODUCT((range-threshold)*(range>threshold))`

Submission Rules

> ⚠️ Save as **.xlsx only**. Sheet names `Lab`, `Assignment`, `Written` must not be renamed, reordered, or deleted. Answers go in **Column C** of the Lab sheet. Do not insert, delete, or move any column.

Filename: `CE310_W6_<NetID>.xlsx` — upload to the D2L Dropbox by **9:00 AM next Wednesday**.